

TABLAS DE DECISIÓN Y PSEUDOCÓDIGO

1 (6 Puntos) Crea la **tabla de decisión** y el **Pseudocódigo** de una aplicación encargada de hacer de pasarela de pago, procesando transacciones de tarjeta de crédito y/o débito.

- Se permite pagar con distintos tipos de tarjetas de distintas compañías MasterCard o Visa pudiendo fraccionarse el pago en dos.
- En caso de fraccionar el pago, las dos tarjetas no pueden ser de débito.

La aplicación preguntará la forma de pago e indicará si está permitida o no.

FRACCIONADO	0	0	0	0	1	1	1	1
CRÉDITO	0	0	1	1	0	0	1	1
DEBITO	0	1	0	1	0	1	0	1
VÁLIDO	0	1	1	0	0	0	1	1

```

Algoritmo sin titulo
    f=falso; //pago fraccionado
    c=verdadero; //tarjeta de crédito
    d=verdadero; //tarjeta de débito
    si (!f&!c&!d) Entonces
        escribir "DENEGADO"
    FinSi
    si (!f&!c&d) Entonces
        escribir "ACEPTADO"
    FinSi
    si (!f&c&!d) Entonces
        escribir "ACEPTADO"
    FinSi
    si (!f&c&d) Entonces
        escribir "DENEGADO"
    FinSi
    si (f&!c&!d) Entonces
        escribir "DENEGADO"
    FinSi
    si (f&!c&d) Entonces
        escribir "DENEGADO"
    FinSi
    si (f&c&!d) Entonces
        escribir "ACEPTADO"
    FinSi
    si (f&c&d) Entonces
        escribir "ACEPTADO"
    FinSi
FinAlgoritmo

```

2 (4 Puntos) Se pretende crear una aplicación para grabar películas en DVD's de 4,7GBytes realizando el mejor ajuste para no desaprovechar el espacio de almacenamiento.

Las películas están catalogadas por distintos géneros, ACCIÓN, TERROR, COMEDIA y SUSPENSE. En un mismo DVD no se podrán mezclar películas de COMEDIA y TERROR.

Tampoco se quiere poner en el mismo DVD las películas Jugada salvaje y El cuerpo ni Furia de Titanes y El hombre de acero.

	TÍTULO	GÉNERO	TAMAÑO EN GB
1	La última casa a la izquierda	TERROR	1,830
2	Saw IV	TERROR	1,435
3	La huérfana	TERROR	2,163
4	Furia de Titanes	ACCIÓN	1,746
5	El hombre de acero	ACCIÓN	0,964
6	Los Vengadores	ACCIÓN	2,032
7	American Pie: El reencuentro	COMEDIA	1,257
8	El lado bueno de las cosas	COMEDIA	3,139
9	Los tres chiflados	COMEDIA	0,750

10	Jugada salvaje	SUSPENSE	2,275
11	El cuerpo	SUSPENSE	2,082
12	15 años y un día	SUSPENSE	2,321

Observamos que cada vez que se grabe un disco quedarán menos películas y por lo tanto menos combinaciones posibles.

Para resolver el primer caso podemos aprovechar el siguiente código que genera las 8192 posibles combinaciones con 12 bits.

```


Algoritmo sin_titulo
  definir t Como Entero
  Dimension bit(13)
  para t<-2 hasta 8191 hacer
    para k<-1 hasta 12 hacer //inicializamos los 12 bits a 0
      bit[k]=0
    FinPara
    dec<-t
    bin<-0
    i<-0
    Repetir
      r<-dec mod 2
      bin<-bin+r*(10^i)
      i<-i+1
      bit[i]=r
      dec<-trunc(dec/2)
    Hasta Que dec=1
    bit[i+1]=dec
    bin<-bin+10^i
    Mostrar bit[12],bit[11],bit[10],bit[9],bit[8],bit[7],bit[6],bit[5],bit[4],bit[3],bit[2],bit[1]
  fin para
FinAlgoritmo

```

Modifica el Pseudocódigo para que seleccione las películas que mejor se ajusten al tamaño del DVD cumpliendo las restricciones impuestas.

La solución es:

`tam=4.7 //tamaño del disco`



El código:

```

Algoritmo sin_titulo
  definir t Como Entero
  tam=4.7 //tamaño del disco
  Dimension bit(12)
  Dimension sol(12)
  Dimension pelicula(12)
  pelicula[1]="La última casa a la izquierda --- TERROR ---- 1,830 GB"
  pelicula[2]="Saw IV ----- TERROR ---- 1,435 GB"
  pelicula[3]="La huérfana ----- TERROR ---- 2,163 GB"
  pelicula[4]="Furia de Titanes ----- ACCIÓN ---- 1,746 GB"
  pelicula[5]="El hombre de acero ----- ACCIÓN ---- 0,964 GB"
  pelicula[6]="Los Vengadores ----- ACCIÓN ---- 2,032 GB"
  pelicula[7]="American Pie: El reencuentro ---- COMEDIA ---- 1,257 GB"
  pelicula[8]="El lado bueno de las cosas ----- COMEDIA ---- 3,139 GB"
  pelicula[9]="Los tres chiflados ----- COMEDIA ---- 0,750 GB"
  pelicula[10]="Jugada salvaje ----- SUSPENSE --- 2,275 GB"
  pelicula[11]="El cuerpo ----- SUSPENSE --- 2,082 GB"
  pelicula[12]="15 años y un día ----- SUSPENSE --- 2,321 GB"
  // no contemplamos los dos primeros casos
  // ninguna película y solo una

```

```


minimo=tam;
para t<-2 hasta 8191 hacer
  para k<-1 hasta 12 hacer //inicializamos los 12 bits a 0
    bit[k]=0
  FinPara
  dec<-t
  bin<-0
  i<-0
  Repetir
    r<-dec mod 2
    bin<-bin+r*(10^i)
    i<-i+1
    bit[i]=r
    dec<-trunc(dec/2)
  Hasta Que dec=1
  bin<-bin+10^i
  size= bit[12]*2.321+bit[11]*2.082+bit[10]*2.275+bit[9]*0.750+bit[8]*3.139+bit[7]*1.257;
  size= size+bit[6]*2.032+bit[5]*0.964+bit[4]*1.746+bit[3]*2.163+bit[2]*1.435+bit[1]*1.830;
  //restricciones no puede haber peliculas de TERROR y COMEDIA en el mismo disco
  si (!((bit[1]==1) o (bit[2]==1) o (bit[3]==1)) y ((bit[7]==1) o (bit[8]==1) o (bit[9]==1))) entonces

    // Si el tamaño se ajusta mejor a 4,7GBytes
    si ((tam-size)>=0) y ((tam-size)<minimo) entonces
      minimo=tam-size;
      //guardamos la combinación de películas más cercana a 4,7GBytes
      para w<-1 hasta 12 hacer
        si (bit[w]==1) entonces
          sol[w]=verdadero
        Sino
          sol[w]=falso
        FinSi
      fin para
    FinSi
  fin para
  //Muestra la solución
  para t<-1 hasta 12 hacer
    si sol[t] entonces
      Mostrar pelicula[t]
    FinSi
  FinPara
  Mostrar "TAMAÑO TOTAL:", tam-minimo;
FinAlgoritmo

```

Probando con distintos tamaños de disco observamos que el algoritmo encuentra el mejor ajuste cumpliendo la restricción de no incluir en el mismo disco películas de terror y comedias:

tam=8.5 //tamaño del disco




```

*** Ejecución Iniciada. ***
Saw IV ----- TERROR ----- 1,435 GB
Furia de Titanes ----- ACCIÓN ----- 1,746 GB
El hombre de acero ----- ACCIÓN ----- 0,964 GB
Los Vengadores ----- ACCIÓN ----- 2,032 GB
15 años y un día ----- SUSPENSE --- 2,321 GB
TAMAÑO TOTAL:8.498
*** Ejecución Finalizada. ***

```

tam=13.3 //tamaño del disco




```

*** Ejecución Iniciada. ***
Furia de Titanes ----- ACCIÓN ----- 1,746 GB
Los Vengadores ----- ACCIÓN ----- 2,032 GB
American Pie: El reencuentro ----- COMEDIA --- 1,257 GB
El lado bueno de las cosas ----- COMEDIA --- 3,139 GB
Los tres chiflados ----- COMEDIA --- 0,750 GB
Jugada salvaje ----- SUSPENSE --- 2,275 GB
El cuerpo ----- SUSPENSE --- 2,082 GB
TAMAÑO TOTAL:13.281
*** Ejecución Finalizada. ***

```

```
tam=15.9 //tamaño del disco
```



```

*** Ejecución Iniciada. ***
La última casa a la izquierda --- TERROR ----- 1,830 GB
Saw IV ----- TERROR ----- 1,435 GB
La huérfana ----- TERROR ----- 2,163 GB
Furia de Titanes ----- ACCIÓN ----- 1,746 GB
Los Vengadores ----- ACCIÓN ----- 2,032 GB
Jugada salvaje ----- SUSPENSE ----- 2,275 GB
El cuerpo ----- SUSPENSE ----- 2,082 GB
15 años y un día ----- SUSPENSE ----- 2,321 GB
TAMAÑO TOTAL:15.884
*** Ejecución Finalizada. ***

```

Nos falta poner la restricción:

“Tampoco se quiere poner en el mismo DVD las películas Jugada salvaje y El cuerpo ni Furia de Titanes y El hombre de acero.”

Esto lo conseguimos con la condición:

```
si !(((bit[10]==1) y (bit[11]==1)) o ((bit[4]==1) y (bit[5]==1))) entonces
```


Si observamos la salida para los tamaños de disco:

8.5 GB -> Aparecen: *Furia de Titanes* y *El hombre de acero*.

13.3 y 15.9 GB -> -> Aparecen: *Jugada salvaje* y *El cuerpo*.

Añadiendo la restricción al código las salidas para esos tamaños ya no incluyen las películas en conflicto:

```
tam=8.5 //tamaño del disco
```




```

*** Ejecución Iniciada. ***
El lado bueno de las cosas ----- COMEDIA ----- 3,139 GB
Los tres chiflados ----- COMEDIA ----- 0,750 GB
Jugada salvaje ----- SUSPENSE ----- 2,275 GB
15 años y un día ----- SUSPENSE ----- 2,321 GB
TAMAÑO TOTAL:8.485
*** Ejecución Finalizada. ***

```

```
tam=13.3 //tamaño del disco
```



```

*** Ejecución Iniciada. ***
La última casa a la izquierda --- TERROR ----- 1,830 GB
Saw IV ----- TERROR ----- 1,435 GB
La huérfana ----- TERROR ----- 2,163 GB
El hombre de acero ----- ACCIÓN ----- 0,964 GB
Los Vengadores ----- ACCIÓN ----- 2,032 GB
Jugada salvaje ----- SUSPENSE ----- 2,275 GB
15 años y un día ----- SUSPENSE ----- 2,321 GB
TAMAÑO TOTAL:13.02
*** Ejecución Finalizada. ***

```

Código Fuente seleccionable:

Algoritmo sin_titulo

```

    definir t Como Entero
    tam=13.3 //tamaño del disco
    Dimension bit(12)
    Dimension sol(12)
    Dimension pelicula(12)

    pelicula[1]="La última casa a la izquierda --- TERROR ----- 1,830 GB"
    pelicula[2]="Saw IV ----- TERROR ----- 1,435 GB"
    pelicula[3]="La huérfana ----- TERROR ----- 2,163 GB"
    pelicula[4]="Furia de Titanes ----- ACCIÓN ----- 1,746 GB"
    pelicula[5]="El hombre de acero ----- ACCIÓN ----- 0,964 GB"
    pelicula[6]="Los Vengadores ----- ACCIÓN ----- 2,032 GB"
    pelicula[7]="American Pie: El reencuentro ---- COMEDIA ---- 1,257 GB"
    pelicula[8]="El lado bueno de las cosas ----- COMEDIA ---- 3,139 GB"
    pelicula[9]="Los tres chiflados ----- COMEDIA ---- 0,750 GB"
    pelicula[10]="Jugada salvaje ----- SUSPENSE --- 2,275 GB"
    pelicula[11]="El cuerpo ----- SUSPENSE --- 2,082 GB"
    pelicula[12]="15 años y un día ----- SUSPENSE --- 2,321 GB"
    // no contemplamos los dos primeros casos
    // ninguna película y solo una
    minimo=tam;
    para t<-2 hasta 8191 hacer
        para k<-1 hasta 12 hacer //inicializamos los 12 bits a 0
            bit[k]=0

        FinPara
        dec<-t
        bin<-0
        i<-0
        Repetir
            r<-dec mod 2
            bin<-bin+r*(10^i)
            i<-i+1
            bit[i]=r
            dec<-trunc(dec/2)

        Hasta Que dec=1
        bin<-bin+10^i
        size= bit[12]*2.321+bit[11]*2.082+bit[10]*2.275+bit[9]*0.750+bit[8]*3.139+bit[7]*1.257;
        size= size+bit[6]*2.032+bit[5]*0.964+bit[4]*1.746+bit[3]*2.163+bit[2]*1.435+bit[1]*1.830;
        //restricciones no puede haber películas de TERROR y COMEDIA en el mismo disco
        si !(((bit[1]==1) o (bit[2]==1) o (bit[3]==1)) y ((bit[7]==1) o (bit[8]==1) o (bit[9]==1))) entonces
            //Las películas Jugada salvaje y El cuerpo no pueden estar en el mismo disco.
            // Las películas Furia de Titanes y El hombre de acero no pueden estar en el mismo disco.
            si !(((bit[10]==1) y (bit[11]==1)) o ((bit[4]==1) y (bit[5]==1))) entonces
                // Si el tamaño se ajusta mejor a tam (4,7GBytes)
                si ((tam-size)>=0) y ((tam-size)<minimo) entonces
                    minimo=tam-size;
                    //guardamos la combinación de películas más cercana a tam (4,7GBytes)
                    para w<-1 hasta 12 hacer
                        si (bit[w]==1) entonces
                            sol[w]=verdadero
                        Sino
                            sol[w]=falso
                        FinSi
                    fin para
                FinSi
            FinSi
        FinSi
    fin para
    //Muestra la solución
    para t<-1 hasta 12 hacer
        si sol[t] entonces
            Mostrar pelicula[t]
        FinSi
    FinPara
    Mostrar "TAMAÑO TOTAL:",tam-minimo;
FinAlgoritmo

```